4. Gyakorlat

Maple ismerkedés

Grafikonok, függvényábrázolás I.: a `plot` parancs

x = ..., y = ..., axes, scaling, discont, color, thickness, symbol, symbolsize, style=point, caption, filled=[color=..., transparency=...], font, gridlines=true, labels=[x, y], legend, linestyle, title

```
> restart;
> plot(4 - x^2, x);
> plot(4 - x^2, x = -4..4);
> plot(4 - x^2, x = -4..4, y = -6..6);
> plot(4 - x^2, x = -4..4, y = -6..6, axes = boxed);
> plot(4 - x^2, x = -4..4, y = -6..6, scaling = constrained);
> plot(4 - x^2, x = -4..4, color = blue);
> plot(4 - x^2, x = -4..4, color = blue, thickness = 3);
> plot(4 - x^2, x = -4..4, color = blue, thickness = 3, filled =
  true);
> plot(4 - x^2, x = -4..4, color = blue, thickness = 3, filled =
  [color = red, transparency = 0.5]);
> plot(4 - x^2, x = -4..4, color = blue, thickness = 3, gridlines
  = true);
> plot(4 - x^2, x = -4..4, color = blue, thickness = 3, linestyle
  = dash);
> plot(4 - x^2, x = -4..4, color = blue, thickness = 3, title="Ez
  egy függvény");
> plot(4 - x^2, x = -4..4, color = blue, thickness = 3, title =
  "Ez egy függvény", caption = "Egy parabolát ábrázolunk plot
  opciókkal");
> plot(4 - x^2, x = -4..4, color = blue, thickness = 3, title =
  "Ez egy függvény", caption = "Egy parabolát ábrázolunk plot
  opciókkal", legend="Parabola1");
> plot([[1, 1], [2, 1.5], [3, 2], [4, 3], [5, 4]], x = 0 .. 5, y
  = 0..5, style = point);

> plot([1, 2, 3, 4, 5], [1, 1.5, 2, 3, 4], x = 0..5, y = 0..5,
  style = point, symbol = cross, symbolsize = 20);
> f := x -> piecewise(x <= 0, -x, x <= 1, 1 - x, x <= 2, 2, 3);
  'f(x)' = f(x);
> plot(f(x), x = -2..4, discont = true, color = blue, thickness =
  4, title = "Nem folytonos függvény", symbolsize = 30, symbol =
  solidcircle);
> plot(f(x), x = -2..4, discont = true, title = "Nem folytonos
  függvény", font = ["HELVETICA", bold, 12], symbol =
  solidcircle);
> plot([sin(t), cos(t)/t, t = 0..Pi/2], x = -1..2, y = -1..10);
> plot3d(sqrt(x^2 + y^2), x = -1..1, y = -1..1);
```

Feltételes valószínség, események függetlensége

A valószínség egy olyan mérték a valószínségi mez eseményterében, mely szerint a teljes eseménytér mértéke 1. Viszont még egy egyébként helyes mérték is csak akkor mutatja az egyes események valószínségét a kísérletekkel összhangban, ha a kísérletek valóban véletlenszerek a teljes eseménytéren. Elfordulhat azonban, hogy a kísérlet során olyan körülmény merül fel, mely azáltal befolyásolja a kísérletet, hogy (könnyebben vagy nehezebben kideríthet módon) az eshetségek valódi számát (az eseményteret) szkíti, bizonyos lehetségeket eleve kizár, és így a kísérleti tapasztalat nincs összhangban az elméleti valószínséggel. Akár éppen ez utalhat a kísérlet tervezésének vagy az elméleti modellnek a hibájára. Ilyenkor a számítás korrigálható azzal, hogy az elméleti eseményteret leszkítjük, és a valószínségeket ezen leszkített eseménytér mértékéhez arányítjuk.

Például a Földön az emberek $\frac{1}{5}$ része kínai. Ez viszont nem jelenti, hogy az utcán mondjuk egy év alatt velünk szembe jöv emberek nagyjából ötöde kínai, mert ez függ a kísérlet kivitelezésétl: más eredményt kapunk, ha a kísérletet Pécsen avagy Pekingben végezzük el. Egyikben sem fog kijönni az elméleti 20 $\%$ körüli arány - Pécsen jóval alatta, Pekingben jóval felette lesz. Az említett kísérletek ugyanis valójában nem abban az eseménytérben (az összes ember halmazán) zajlanak, melyben a kínaiak aránya (elfordulási valószínsége) 20 $\%$, hanem egy-egy ersen leszkített eseménytéren: a Pécs ill. Peking térségében elforduló emberek halmazán. A kísérlet eredményei akkor jönnek összhangba a modellel, ha az eseményteret (összes ember a Földön) leszkítjük a vizsgált város populációjára, és a valószínségeket ennek megfelelen újrafogalmazzuk: a kínaiak elfordulására más-más (feltételes) valószínségekkel számolunk attól a feltételtl függen, hogy a vizsgálatot Pécsen vagy Pekingben végezzük.
Különféle (véletlenszerséget és logikát egyszerre alkalmazó) játékokban (és persze az élet véresen komoly dolgaiban is) fontos a menet közben befolyó információk nyomán a valószínségi modell –gyakran az effektív eseménytér, a feltételes valószínűség– folyamatos korrekciója. Tipikus példák a kártyajátékok, melyek arra épülnek, hogy a játékosok nem rendelkeznek a helyzetrl teljes információval (nem ismerik a többiek lapjait). Ugyanakkor egy tapasztalt játékos jól érez ezzel kapcsolatban bizonyos valószínségeket, melyekre stratégiát építhet. Az igazán jó játékos ezen túl még arra is képes, hogy nem csupán a lehetségek átlagos valószínségét érzi, hanem egy-egy konkrét játék menete során elkerül (a nála lév, vagy a "kiment", vagy mások által felfedett) lapok megfigyelésével folyamatosan nyomon követi az eseménytér szkülését (pl. "három ász kiment, egy nálam van, másnál már nem lehet"): a valódi helyzethez jobban illeszked valószínségi modellel, aktuális feltételes valószínségekkel számol, így statisztikailag hosszabb távon elnyre tesz szert.
Ezen túl a feltételes valószínségre épül események függetlenségének a vizsgálata: függ-e egy esemény valószínsége egy másik bekövetkezésétl vagy sem.

Kidolgozott példafeladatok

1. Feladat (Kockadobások eseményei, feltételes valószínség, függetlenség)

Dobjunk egy hatoldalú kockával. Jelölje A a 'páros dobás', B pedig a 'legalább 4-es dobás' eseményét. Határozzuk meg az A, a B és az $A \cdot B$ események valószínségét! Mekkora eséllyel dobtunk párosat, feltéve, hogy legalább 4-est dobtunk? Független-e az A és a B esemény egymástól?

Megoldás

```
> restart;
```
Az eseménytér a lehetséges dobások halmaza: $\Omega = \{1, 2, 3, 4, 5, 6\}$.
Tekintsük a 'páros dobás' és a 'legalább 4-es dobás' eseményeit:
$A = \{2, 4, 6\}$, $B = \{4, 5, 6\}$ Ezek valószínségi mértékei a teljes eseménytérben:
```
> pA := 3/6;
  pB := 3/6;
```
Ha legalább 4-et és párosat dobtunk, akkor vagy 4-est vagy 6-ost dobtunk, azaz $A \cdot B = \{4, 6\}$. Ebbl az $A \cdot B$ esemény valószínsége:
```
> pAB := 2/6;
```
A 'páros dobás' eseményének a 'legalább 4-es dobás' eseményre vonatkoztatott feltételes valószínsége alatt azt értjük, hogy utóbbi bekövetkezte mellett mekkora valószínség az elbbi. Ez a feltételes valószínség tartalma és definíciója szerint a két esemény együttes bekövetkezésének valószínsége a feltételként megfogalmazott esemény valószínségéhez viszonyítva. Lényegében arról van szó, hogy az eseményteret a feltétel-eseményre szkítjük, és ehhez arányítjuk a különféle események ezen belülre es részének mértékét.
```
> `P(A|B)` := pAB/pB;
```
Ez az érték arra utal, hogy a 'legalább 4' értékek kétharmad része páros.
Ha egy A eseménynek valamely B eseményre vonatkoztatott valószínsége megegyezik az A esemény valószínségével (vagyis a B esemény bekövetkezése nincs hatással az A esemény bekövetkezésének valószínségére), akkor azt mondjuk, hogy az A esemény független a B eseménytl. Bár a definíció önmagában aszimmetrikus (megkülönbözteti az épp megfigyelt eseményt attól a referencia-eseménytől, melyre –annak bekövetkeztét feltételezve– vonatkoztatunk), be lehet bizonyítani, hogy maga a reláció szimmetrikus, vagyis kölcsönös értelemben lehet beszélni két esemény függetlenségérl vagy függségérl (A pontosan akkor független B-tl, ha B független A-tól).

Független-e az A és a B események?
Miután láttuk, hogy az eredetileg $\frac{1}{2}$ valószínség 'páros dobás' valószínsége a 'legalább 4-es dobás'-ra vonatkoztatva $\frac{2}{3}$-ra változik, vagyis utóbbi bekövetkezése befolyásolja elbbi valószínségét, természetesen nem függetlenek. Szemléletesen is érthet, hogy a 'legalább 4-es' dobások között több a páros, mint a páratlan, tehát ezen belül nézve nagyobb a 'páros dobás' valószínsége. A kapcsolat nem mindig ilyen egyszer és szemléletes, de a definíció alapján megbízhatóan és egyszeren megállapítható.

Szimuláció

Végezzünk szimulációt 1000 kockabobással és ábrázoljuk grafikonon a 'páros dobás' relatív gyakoriságát a 'legalább 4-es dobás' feltétele mellett!
```
> with(Statistics):
```
Készítsünk egy diszkrét valószínségi változót a kockadobás modellezésére!
```
> n := 1000;
  randomize();
  X := RandomVariable(DiscreteUniform(1,6));
```
Végezzük el az 1000 kockadobást!
```
> dobasok := Sample(X, n):
```
Ezután menjünk végig a dobások listáján, és gyjtsük két vektorba a legalább 4-es dobások gyakoriságát és a legalább 4-es páros dobások gyakoriságát:
```
> vAB, vB := Vector(n), Vector(n):
```

```
freqAB, freqB := 0, 0:
```

```
> for i from 1 to n do
    dob := round(dobasok[i]):
    if dob >= 4 then
      freqB := freqB + 1:
      if dob mod 2 = 0 then
        freqAB := freqAB + 1:
      end if:
    end if:
    vAB[i] := freqAB:
    vB[i] := freqB:
  end do:
```

Számoljuk ki a feltételes relatív gyakoriságok sorozatát és tegyük bele egy vektorba:

```
> rFreqV := Vector(n):
  for i from 1 to n do
    if vB[i] = 0 then
      rFreqV[i] := 0: # amíg a B esemény nem következik
  be, a feltételes relatív gyakoriságot vegyük 0-nak
    else
      rFreqV[i] := vAB[i]/vB[i]:
    end if:
  end do:
```

Végül ábrázoljuk grafikonon a feltételes relatív gyakoriságokat a Statistics csomag LineChart parancsával! A fentebb kiszámított feltételes valószínségnél húzzunk vízszintes vonalat az összehasonlítás kedvéért.

```
> P1 := LineChart(rFreqV, gridlines = true, axes = boxed,
  symbolsize = 3):
  P2 := plot(`P(A|B)`, x = 1..n, y = 0..1):
  plots[display](P1, P2);
```

Meggyzen látszik, hogy a kísérletek számának növelésével a feltételes relatív gyakoriságok az elméleti feltételes valószínséghez $\left(\frac{2}{3}\right)$ tartanak:

```
> evalf(rFreqV[n]);
```

2. Feladat (Feltételes valószínség, függetlenség táblázatban)

Két szabályos 6 oldalú dobókockával dobunk. Jelölje A azt az eseményt, hogy az összeg 4 vagy 5 vagy 6 lesz, míg B azt, hogy mindkét dobás páros!

a) Számítsuk ki a $P(A|B)$ feltételes valószínséget!

b) Függetlenek-e egymástól az A és a B események?

c) Hogyan változik az eredmény, ha valamelyik esemény tagadását vesszük?

d) Ábrázoljuk 2×2-es táblázatban Ω elemeinek A és B halmazokba való szétosztását! Mit jelent a függetlenség a táblázatban!

Megoldás

```
> restart;
```

(a) Számítsuk ki a $P(A|B)$ feltételes valószínséget!

Megadjuk az Ω eseménytér elemeit kételem listákkal. Az els elem jelenti az els dobást, a második a másodikat!

```
> M := Matrix(6, 6, (i, j) -> [i, j]);
```

```
> Omega := convert(M, set);
```

```
> nOmega := numelems(Omega); # nops(Omega) is mködik
```
Az

$$A = \{ (a, b) \in E \mid a + b = 4 \textit{ vagy } 5 \textit{ vagy } 6 \}$$

esemény elemeit a *select* parancs segítségével (vagy kézzel felsorolva) adhatjuk meg!

```
> A := select(par -> par[1] + par[2] in {4, 5, 6}, Omega);
> nA := numelems(A);
> pA := nA/nOmega;
```
A

$$B = \{ (a, b) \in E \mid a, b \in \{2, 4, 6\} \}$$

esemény elemeit két egymásba ágyazott sorozattal készítjük el!

```
> B := {seq(seq([i, j], j = 2..6, 2), i = 2..6, 2)};
> nB := numelems(B);
> pB := nB/nOmega;
```
Az A és B halmazok $A \cap B$ metszetét a Maple *intersect* mveletével kapjuk!

```
> AB := A intersect B;
> nAB := numelems(AB);
> pAB := nAB/nOmega;
```
Az A esemény B eseményre vonatkozó feltételes valószínségét a

$$P(A|B) = \frac{P(A \cdot B)}{P(B)}$$

képlettel számoljuk definíció szerint!

```
> `P(A|B)` := pAB/pB;
```
Tehát az A esemény feltételes valószínsége a B eseményre vonatkozóan $\frac{1}{3}$.

(b) Függetlenek-e egymástól az *A* és a *B* események?

A függetlenség egymással ekvivalens két feltételét is ellenrízzük:

$$P(A|B) = P(A)$$
$$P(A \cdot B) = P(A) \cdot P(B).$$

```
> ellenorzes1 := `P(A|B)` = pA;
  evalb(%); # az ellenrzés igazságértéke
> ellenorzes2 := pAB = pA*pB;
  evalb(%);
```
Tehát az A és a B események függetlenek valószínségi (sztochasztikus) értelemben.

(c) Hogyan változik az eredmény, ha valamelyik esemény tagadását vesszük?

```
> Ac := Omega minus A; # A tagadása (ellentét eseménye,
  komplementere)
> nAc := numelems(Ac); # = nOmega - nA
> pAc := nAc/nOmega;
```
Ellenrizhetjük, hogy ez megegyezik $1 - P(A)$-val:

```
> evalb(pAc = 1 - pA);
> AcB := Ac intersect B;
> nAcB := numelems(AcB);
> pAcB := nAcB/nOmega;
> ellenorzes3 := pAcB = pAc*pB;
  evalb(%);
```
Tehát az $\bar{A}$ esemény független a B eseménytl!

```
> Bc := Omega minus B;
> nBc := numelems(Bc);
```

```
> pBc := nBc/nOmega;
```
Ellenrizhetjük, hogy ez megegyezik $1 - P(B)$-vel:
```
> evalb(pBc = 1 - pB);
> ABc := A intersect Bc;
> nABc := numelems(ABc);
> pABc := nABc/nOmega;
> ellenorzes4 := pABc = pA*pBc;
  evalb(%);
```
Tehát az A esemény független a $\overline{B}$ eseménytl!
```
> AcBc := Ac intersect Bc;
> nAcBc := numelems(AcBc);
> pAcBc := nAcBc/nOmega;
> ellenorzes5 := pAcBc = pAc*pBc;
  evalb(%);
```
Tehát az $\overline{A}$ esemény független a $\overline{B}$ eseménytl!
A példán kersztül láttuk, hogy

(i) A és B események függetlenek;
(ii) $\overline{A}$ és B is függetlenek;
(iii) A és $\overline{B}$ is függetlenek;
(iv) $\overline{A}$ és $\overline{B}$ is függetlenek.

Általánosan igaz a következ:

TÉTEL. Ha A és B független események, akkor
(1) $\overline{A}$ és B is függetlenek;
(2) A és $\overline{B}$ is függetlenek;
(3) $\overline{A}$ és $\overline{B}$ is függetlenek.

(d) Ábrázoljuk 2×2-es táblázatban Ω elemeinek A és B halmazokba való szétosztását! Mit jelent a függetlenség a táblázatban!
Beszúrunk egy "spreadsheet" táblázatot a beilleszt men "táblázat" alpontjával!

Események						
	A	B	C	D	E	F
1	0	*B*	*Bc*	*összeg*		
2	*A*	3	9	12		
3	*Ac*	6	18	24		
4	*összeg*	9	27	36		
5						
6						
7						
8						
9						
10						

A függetlenség itt a táblázatban a következ egyenlségek teljesülését jelenti: a táblázat bármely bels (citromsárga) cellájának értéke egyenl a sora és az oszlopa végén lev összegek (narancssárga) szorzata, osztva a jobb alsó sarokban lev totál összeggel (rózsaszín)!

```
> evalb(3 = (12*9)/36);
  evalb(9 = (12*27)/36);
  evalb(6 = (24*9)/36);
  evalb(18 = (24*27)/36);
```

Szimuláció

Végezzünk szimulációt 10000 kockabobással és ábrázoljuk grafikonon az *A* esemény relatív gyakoriságát a *B* esemény bekövetkezésének feltétele mellett!

```
> with(Statistics):
```

Készítsünk két diszkrét valószínségi változót a két kockával való dobás modellezésére!

```
> n := 10000;
  randomize();
  X := RandomVariable(DiscreteUniform(1,6));
  Y := RandomVariable(DiscreteUniform(1,6));
```

Végezzük el az 1000 kockadobást!

```
> dobasokX := Sample(X, n):
  dobasokY := Sample(Y, n):
  dobasok := [seq([round(dobasokX[i]), round(dobasokY[i])
  ], i = 1..n)]:
```

Készítsünk segédeljárásokat az *A* és *B* események bekövetkezésének ellenrzésére!

```
> esemenyA := proc(x, y)
    return evalb(x + y in {4, 5, 6});
  end proc:
> esemenyB := proc(x, y)
    return evalb((x mod 2 = 0) and (y mod 2 = 0));
```

```
end proc:
```

Ezután menjünk végig a dobás-párok listáján, és gyjtsük két vektorba a B és $A \cdot B$ események bekövetkezési gyakoriságát:

```
> vAB, vB := Vector(n), Vector(n):
  freqAB, freqB := 0, 0:
> for i from 1 to n do
    dobX := dobasok[i][1]:
    dobY := dobasok[i][2]:
    if esemenyB(dobX, dobY) then
      freqB := freqB + 1:
      if esemenyA(dobX, dobY) then
        freqAB := freqAB + 1:
      end if:
    end if:
    vAB[i] := freqAB:
    vB[i] := freqB:
  end do:
```

Számoljuk ki a feltételes relatív gyakoriságok sorozatát és tegyük bele egy vektorba:

```
> rFreqV := Vector(n):
  for i from 1 to n do
    if vB[i] = 0 then
      rFreqV[i] := 0: # amíg a B esemény nem következik
  be, a feltételes relatív gyakoriságot tekintsük 0-nak
    else
      rFreqV[i] := vAB[i]/vB[i]:
    end if:
  end do:
```

Végül ábrázoljuk grafikonon a feltételes relatív gyakoriságokat a Statistics csomag LineChart parancsával! A fentebb kiszámított feltételes valószínségnél húzzunk vízszintes vonalat az összehasonlítás kedvéért.

```
> P1 := LineChart(rFreqV, gridlines = true, axes = boxed,
  symbolsize = 3):
  P2 := plot(`P(A|B)`, x = 1..n, y = 0..1):
  plots[display](P1, P2);
```

Meggyzen látszik, hogy a kísérletek számának növelésével a feltételes relatív gyakoriságok az elméleti feltételes valószínséghez $\left(\frac{1}{3}\right)$ tartanak:

```
> evalf(rFreqV[n]);
```

3. Feladat (Sorosan és párhuzamosan kapcsolt rendszerek megbízhatósága)

Az alábbi áramkör csak akkor mködik, ha balról jobbra haladva van olyan útvonal, amelyben mköd alkatrészek vannak. Az egyes alkatrészek megbízhatóságát az ábrán látjuk.



Az alkatrészek meghibásodása egymástól független. Mekkora a valószínsége, hogy mködik a teljes áramkör?

Megoldás

```
> restart;
```

Az ábrán látható A, B és C blokkok kett-kett független, *párhuzamosan* kapcsolt alkatrészbl állnak! Egy ilyen blokk csak akkor nem mködik, ha egyik alkatrészük sem mködik, ezért a párhuzamosan kapcsolt rendszer ered megbízhatóságára vonatkozó számítást kell használni, vagyis az egybl kivont megbízhatóságok szorzatát ki kell vonni 1-bl.

```
> p1, p2, p3, p4, p5, p6 := 0.9, 0.95, 0.9, 0.95, 0.8,
  0.9;
```

```
> pA := 1 - (1 - p1)*(1 - p2); # 1 - p1: az 1. alkatrész
  nem mködik, 1 - p2: az 2. alkatrész nem mködik, (1 -
  p1)*(1 - p2): sem az 1., sem a 2. alkatrész nem mködik
```

Látható, hogy a párhuzamos kapcsolás javítja a megbízhatóságot, új alkatrész beépítése révén! Hasonlóan, a B és C blokkok megbízhatósága:

```
> pB := 1 - (1 - p3)*(1 - p4);
> pC := 1 - (1 - p5)*(1 - p6);
```

Az A, B és C független blokkok egymáshoz *sorosan* kapcsoltak, ezért ezek ered megbízhatóságát az egyes megbízhatóságok szorzata adja!

```
> p := pA*pB*pC;
```

Tehát az áramkör ered megbízhatósága kb. 97 %-os.

4. Feladat (Mveletek eseményekkel és valószínségeik)

Határozzuk meg a $P(A)$, $P(B)$ és $P(A\cdot B)$ valószínségeket, ha adottak $P(A|B) = \frac{1}{5}$, $P(A|\overline{B}) = \frac{1}{3}$ és $P(\overline{B}|\overline{A}) = \frac{1}{2}$ feltételes valószínségek! Független-e az A és a B esemény?

Megoldás

```
> restart;
```

A megadott feltételes valószínségeket felvesszük egy-egy változóba a Maple többszörös értékadó utasítását használva!

```
> `P(A|B)`, `P(A|Bc)`, `P(Bc|Ac)` := 1/5, 1/3, 1/2;
```

Írjuk fel a feltételes valószínséget definíció szerint: $P(A|B) = \frac{P(A\cdot B)}{P(B)}$. A Maple a kifejezések teljes kiértékelési folyamatának köszönheten az egyenletben behelyettesíti azoknak a változóknak az értékeit, amelyek adottak!

```
> def1 := `P(A|B)` = pAB/pB;
```

Használva a tagadás esemény és a különbség valószínségére levezetett $P(\overline{B}) = 1 - P(B)$ és $P(A\cdot\overline{B}) = P(A - B) = P(A) - P(A\cdot B)$ azonosságokat,

```
> def2 := `P(A|Bc)` = (pA - pAB)/(1 - pB);
```

Használva a tagadás esemény valószínségére és az összeg esemény valószínségére levezetett $P(\overline{A}) = 1 - P(A)$ és $P(A + B) = P(A) + P(B) - P(A\cdot B)$ azonosságokat,

```
> def3 := `P(Bc|Ac)` = (1 - pA - pB + pAB)/(1 - pA);
```

Mindegyik adott feltételes valószínség kiszámítását visszavezettük a $P(A)$, $P(B)$ és $P(A\cdot B)$ ismeretlenekkel adott formulára. Kaptunk 3 egyenletet 3 ismeretlennel! Oldjuk meg a Maple *solve* eljárásával!

```
> unassign('pA', 'pAB', 'pB'); # töröljük a változók értékét (ha esetleg már korábban lefuttattuk volna ezt a blokkot)
  solve({def1, def2, def3}, {pA, pAB, pB});
  assign(%); # az assign(%) paranccsal hozzá is rendelhetjük a változókhoz a megoldásul kapott értékeket
```

Tehát $P(A) = \frac{3}{11}$, $P(A \cdot B) = \frac{1}{11}$, $P(B) = \frac{5}{11}$. Az A és B függek, mert

$P(A \cdot B) = \frac{1}{11} \neq P(A) \cdot P(B) = \frac{3}{11} \cdot \frac{5}{11}$.

```
> evalb(pA*pB = pAB);
```

5. Feladat (ZH, 2016, A csoport)

Határozzuk meg a $P(A)$, $P(B)$ és $P(A \cdot B)$ valószínségeket, ha adottak a $P(B|A) = \frac{1}{15}$, $P(A+B) = \frac{11}{15}$ és $P\left(A \cdot \overline{B}\right) = \frac{2}{3}$ valószínségek! Függetlenek-e az A és a B események?

Megoldás

```
> restart;
```

A megadott valószínségeket felvesszük egy-egy változóba!

```
> `P(B|A)` := 1/15;
  `P(A + B)` := 11/15;
  pABc := 2/3;
```

Írjuk fel a feltételes valószínséget definíció szerint: $P(B|A) = \frac{P(A \cdot B)}{P(A)}$. A Maple a kifejezések teljes kiértékelési folyamatának köszönheten az egyenletben behelyettesíti azoknak a változóknak az értékeit, amelyek adottak!

```
> def1 := `P(B|A)` = pAB/pA;
```

Használva az összeg valószínségére levezetett
$P(A+B) = P(A) + P(B) - P(A \cdot B)$ azonosságot,

```
> def2 := `P(A + B)` = pA + pB - pAB;
```

Használva a különbség valószínségére levezetett
$P(A \cdot \overline{B}) = P(A - B) = P(A) - P(A \cdot B)$ azonosságot,

```
> def3 := pABc = pA - pAB;
```

Mindegyik adott feltételes valószínség kiszámítását visszavezettük a $P(A)$, $P(B)$ és $P(A \cdot B)$ ismeretlenekkel adott formulára. Kaptunk 3 egyenletet 3 ismeretlennel! Oldjuk meg a Maple *solve* eljárásával!

```
> unassign('pA', 'pAB', 'pB'); # töröljük a változók értékét (ha esetleg már korábban lefuttattuk volna ezt a blokkot)
  solve({def1, def2, def3}, {pA, pAB, pB});
  assign(%); # az assign(%) paranccsal hozzá is rendelhetjük a változókhoz a megoldásul kapott értékeket
```

Tehát $P(A) = \frac{5}{7}$, $P(A \cdot B) = \frac{1}{21}$, $P(B) = \frac{1}{15}$. Az A és B függetlenek, mert

$$P(A \cdot B) = \frac{1}{21} = P(A) \cdot P(B) = \frac{5}{7} \cdot \frac{1}{15}.$$

```
> evalb(pAB = pA*pB);
```

Mindennapi életünk során leginkább szemléletünk kormányoz minket – többé-kevésbé jól (amikor pedig nem, akkor a szemléleten is fejleszteni lehet tanulással). A matematika (azon belül pedig a valószínségelmélet is) ismer olyan nevezetes paradoxonokat (látszólagos ellentmondásokat), melyeknél általában a szemléletünk becsap és helytelen következtetésre vezet minket. A bvészet voltaképp kizárólag ebbl él, az ilyen paradoxonok pedig a matematika bvésztrükkjei. A feltételes valószínség (a szkül eseménytér) téves értelmezésének (figyelmen kívül hagyásának) szép példája a *Monty Hall* paradoxon, mely egy televíziós játékvezet nevét rzi.

6. Feladat (Monty Hall paradoxon)

Egy kvízjáték végén a fnyereményt a következk szerint lehetett megnyerni:
Három függöny egyike mögött rejtve van az értékes fnyeremény (pl. autó). A másik kett nem nyer (állítólag a játékban tréfából kecske volt ezek mögött). A játékos választhat egy függönyt, majd a játékvezet azt mondja, hogy a nem választottak közül kinyitnak egy olyat, ami mögött nincs autó – így is tesznek. Ezek után a játékvezető felajánlja a játékosnak, hogy ha szeretné, módosíthatja döntését: az eredeti választása helyett kérheti a másik, még zárva maradt függönyt. Az a kérdés, hogy érdemes-e a játékosnak változtatni az eredeti választásán?

Megoldás

Többségünk szemlélete azt súgja: nem történt semmi, ami miatt érdemes lenne változtatni, st, az eredetileg $\frac{1}{3}$-nak tn nyerési esély immár 50 %-ra ntt, hiszen a függöny kinyitásával a játék voltaképp újraindult: a nettó helyzet az, hogy az autó a két zárt függöny egyike mögött van, és semmivel sincs rá nagyobb esély, hogy a másik mögött lenne, mint amit eredetileg választottunk (ráadásul elég idegtép lenne, ha éppen a változtatással veszítjük el a fnyereményt). Ehhez képest a meglep igazság az, hogy valószínségelméletileg érdemes elfogadni a váltás lehetségét! A játék ugyanis nem tiszta lappal, az elzményektl függetlenül indul újra a garantáltan nyeretlen függöny kinyitásával.

Matematikai magyarázat

Vezessük be a következ eseményeket:

- A: a játékos eredetileg az autót rejt függönyt választotta;
- K: a játékos eredetileg egy kecskét rejt függönyt választott;
- M: a játékos marad az eredeti választása mellett;
- V: a játékos változtat eredeti döntésén, és a másik függönyt választja
- N: a játékos nyer

A kérdés az, hogy mikor nagyobb a nyerés esélye, ha a játékos marad az eredeti döntésénél, vagy ha változtat rajta – azaz a $P(N|V)$ és a $P(N|M)$ feltételes valószínségek közül melyik nagyobb.

Az eredeti választás szerinti függöny mögött ugyan $P(A) = \frac{1}{3}$ eséllyel ott a fnyeremény, ám ennek kétszeresével, $P(K) = \frac{2}{3}$ eséllyel kecske található mögötte.

1. Eset (a játékos nem változtat). Vegyük észre, hogy ha a játékos végig kitart az eredeti döntése mellett, akkor pontosan akkor nyer, ha eredetileg is a jó függönyt választotta. Tehát $P(N|M)=P(A|M)$.
Az viszont, hogy változtat-e, független attól, hogy eredetileg az autót rejt függönyt választotta-e (hiszen a döntés pillanatában nincs tudatában ennek az információnak, az csak a végén derül ki). Mivel az események függetlensége szimmetrikus reláció, ezért A is független M-tl, így $P(A|M)=P(A)$. Összefoglalva, $P(N|M)=P(A|M)=P(A)$ $= \frac{1}{3}$ a nyerés esélye.

2. Eset (a játékos változtat). Most azt vegyük észre, hogy ha a játékos változtat az eredeti döntésén, akkor pontosan akkor nyer, ha eredetileg egy nem-nyer függönyt választott (mivel a játékvezet garantáltan olyan függönyt nyit ki, ami mögött kecske van, valójában feltálalja a nyereményt a nem választottak között) . Tehát $P(N|V)=P(K|V)$. Az viszont, hogy változtat-e, független attól, hogy eredetileg egy kecskét rejt függönyt választott-e (hiszen a döntés pillanatában nincs tudatában ennek az információnak, az csak a végén derül ki). Mivel az események függetlensége szimmetrikus reláció, ezért K is független V-tl, így $P(K|V)=P(K)$. Összefoglalva, $P(N|V)=P(K|V)=P(K) = \frac{2}{3}$. (Persze az esetek harmadában valóban az elször választott függöny mögött van a nyeremény, és a váltással ilyenkor elveszítjük, kétharmad részben viszont jól járunk vele).
Tehát érdemes változtatni, ugyanis ezzel megduplázzuk nyerési esélyeinket!
Még szemléletesebben látszik a különbség, ha a számarányokon ersebben torzítunk: képzeljük el, hogy 100 függönybl választhatunk 1-et, majd a maradék 99-b1 kinyitnak 98 nem nyert. Ekkor már világosan érezzük, hogy eredeti választásunkkal nem sok esélyünk volt eltalálni a nyereményt, magunk is azt gondoljuk, hogy a maradék 99 között volt, melyek közül végül egyetlen egyet nem nyitnak csak ki - ekkor már elég magától értetd, hogy azt kellene választani.

Egy kis pszichológia
Érdekes lenne megfejteni, hogy a gondolkodási modell vagy akár a lélektan szintjén mi viszi félre a szemléletünket egy ilyesfajta paradoxon esetén - és itt valószínleg szerephez jutnak a tálalás technikai részletei is. Ebben az esetben pl. alighanem fontos a szemlélet megtévesztéséhez, hogy a játékos idben elbb látja a függöny kinyitását, aztán látszólag(!) szabadon, az elzményektl függetlenül választ. Úgy érzi, hogy ha nem lenne mindegy a választás, az olyan lenne, mintha olyan események, mint a függöny kinyitása vagy akár az döntése hatással lennének egy idben korábbi eseményre, a nyeremény elhelyezésére, ami persze ellentétes azzal a szemlélettel, hogy az ok idben mindig megelzi az okozatot. Valójában a nyeremény elhelyezése, majd az els választás van hatással a függöny kinyitására és a második választásra: nem véletlen, hogy melyik függönyt nyitják ki, és hogy melyik marad meg döntési lehetségként. Emellett gondolkodásunk szeret egyszersíteni, nagyon szívesen kiejti a lényegtelennek tn, csupán zavarónak tn tényezket, különösen megfelel tálalásban: ha látványosan (egy ilyen show-ban valószínleg felhajtással: fényekkel, zenével hangsúlyozva) kinyílik egy függöny mögötte a semmivel (pláne kecskével!), akkor úgy érezzük, valami olyan fontos információ birtokába jutottunk (ti. hogy hol biztosan nincs a nyeremény), ami olyan szinten új helyzetet teremt, hogy minden elzménytl függetlenül újra kell értékelni. Ráadásul olyan szép tiszta: kettbl kell választani, egyik mögött ott az autó – ezek után már nehéz pontosan átlátni, milyen

folyamattal is jutottunk el idáig. Ehhez jön még, hogy ezen a ponton úgy tnik, hogy az eredeti választásunkkal a nyerési esély az amúgy is $\frac{1}{3}$ -ról 50 %-ra ugrott, úgyis mindegy: éppen válthatunk is, de ekkor már kötdünk az eredeti választáshoz. Ráadásul, ha épp a váltással veszítenénk, az lélektanilag nagyobb veszteség lenne, mert úgy vesztettük volna el a nyereményt, hogy már megvolt. Aki vált, többnyire az is csak próba-szerencse alapon teszi, nem tudatosul benne, hogy ez ténylegesen növeli az esélyt.

7. Feladat (Színes golyók húzása két dobozból / 1.)

Sárga és kék golyókat helyezünk el vegyesen két dobozba az alábbi ábra szerint:

1. doboz:	2. doboz:
41 sárga 49 kék	9 sárga 1 kék

Eztuán egy golyót sorsolunk ki a következ eljárással: elbb a dobozok közül véletlenszeren kiválasztjuk az egyiket, majd abból véletlenszeren kihúzunk egy golyót.

a) Milyen valószínséggel sorsolunk kék vagy sárga golyót?

b) Készítsük el a dobozok és golyó-színek együttes eloszlás-táblázatát

c) Mekkora a valószínsége, hogy egy sárga golyót az 1. illetve a 2. dobozból húztunk ki?

d) Milyen doboz-választási valószínségek mellett jutna érvényre a sorsolás végeredményében a golyók eredeti számaránya?

Megoldás

```
> restart;
```

A feladat független paraméterei legyenek a következk ($s1$ a sárgák száma az els dobozban stb.):

```
> s1 := 41; k1 := 49; s2 := 9; k2 := 1;
```

Ugyancsak a feladat független paraméterei közé tartozik a valószínség, mellyel az els dobozt választjuk:

```
> pd1 := 1/2;
```

Ebbl következen a másik doboz választásának valószínsége:

```
> pd2 := 1 - pd1;
```

Az összes golyó, majd a sárga, kék ill. az egyes dobozokban lév golyók száma:

```
> g := s1 + k1 + s2 + k2; s := s1 + s2; k := k1 + k2; d1
  := s1 + k1; d2 := s2 + k2;
```

a) Milyen valószínséggel sorsolunk kék vagy sárga golyót?

Noha a rendszerben összességében azonos számban van jelen a kétféle szín golyó, a vázolt kísérleti konstrukcióban egyáltalán nem biztos, hogy egyenl a kisorsolódás valószínsége, mivel szám és arány szerint is egyenetlenül vannak szétosztva a dobozok között. A második dobozban sokkal kevesebb golyó van, viszont a sárgák elsöpr aránybeli fölényben vannak. Ha a két doboz választási esélye azonos, miközben a második az összes golyónak csak kisebb részét tartalmazza, akkor érezzük, hogy az ebben lév golyók (és arányuk) szerepe felértékeldik. Ha els lépésben a választás a második dobozra esik, majdnem biztos, hogy a végül húzott golyó sárga lesz, ezzel szemben az els doboz választása nem valószínsíti hasonló eséllyel a kék választását, hiszen ott majdnem egyenlen vannak (csak kicsivel kevesebb a sárga): az esetek közel felében onnan is sárgát várunk. Szemlélet alapján arra számítunk, hogy lényegesen többször lesz sárga a sorsolás

eredménye, mint kék.
Ebben a kísérleti konstrukcióban a végeredmény színére vonatkozó valószínségi jóslás összetettebb, mivel több ágon futhat végig a folyamat. Könny viszont megfogalmazni a húzott szín valószínűségét egy-egy dobozon belül – vagyis az egyik vagy másik dobozból való színhúzás ún. feltételes valószínségét.

Pl. az els dobozból a kék húzás valószínsége $\frac{49}{90}$. Azt is mondhatjuk, hogy a dobozból húzáskor a 'kék golyó' esemény valószínsége, feltéve, hogy a dobozválasztáskor az 'els doboz' esemény következett be: $\frac{49}{90}$. Vagy ezt: a 'kék golyó' húzásnak az 'els doboz'-ra vonatkoztatott feltételes valószínsége $\frac{49}{90}$.

Ezt így jelölhetjük tömören:
```
> `P(k|d1)` := k1/d1;
```
Hasonlóan definiálhatjuk a többi feltételes valószínséget:
```
> `P(s|d1)` := s1/d1; `P(k|d2)` := k2/d2; `P(s|d2)` :=
  s2/d2;
```
Az adatok összefoglalhatók egy döntési fában:



Minden egyes lefutási ág a kísérlet egy lehetséges kimeneteléhez vezet. Mivel minden kísérlet két lépést jelent (doboz választás, aztán golyó húzás), a lehetséges kimenetelek (elemi események) e két lépés egymásutánjából képezett rendezett párok, melyek összessége az eseménytér: $\Omega = \{(d1, s), (d1, k), (d2, s), (d2, k)\}$. Ezek valószínsége az adott ág szakaszaira írt valószínségek szorzata.
Ezek után a kísérletben az egyes színek kisorsolódási valószínségeit a *teljes valószínség*

tétele (TVT) szerint lehet számolni. Ez lényegét tekintve arról szól, hogy ha egy esemény (itt egy adott szín húzása) több esemény-ágon (különböz dobozokból való húzás) következhet be, melyek ún. teljes eseményrendszert alkotnak, akkor a valószínség az egyes ágakon külön-külön kiszámított valószínségek összegzésével adódik. Így tehát a 'kék' végeredmény valószínsége:

```
> pk := `P(k|d1)`*pd1 + `P(k|d2)`*pd2;
```

A sárgáé pedig:

```
> ps := `P(s|d1)`*pd1 + `P(s|d2)`*pd2;
```

A sárga és kék végeredmény valószínségeinek összege természetesen 1-et ad:

```
> pk + ps;
```

Valóban látszik, hogy az esetek kicsit több mint $\frac{2}{3}$ részében sárga golyó sorsolódik ki, annak ellenére, hogy a rendszer egészében a két szín egyenl arányban van jelen. A kísérletben (a számarányához képest) jelentsen felértékeldött második doboz sárga javára szóló arányai tolták el a kísérlet végeredményét ebbe az irányba.

b) Készítsük el a dobozok és golyó-színek együttes eloszlás-táblázatát

A példa adatai táblázatban is összefoglalhatók (ez a dobozok és színek választásának ún. együttes eloszlása):

Események együttes eloszlása

	A	B	C	D	E	F
1		*S*	*K*	*Össz.*		
2	*D1*	0.2277777778	0.2722222222	0.5000000000		
3	*D2*	0.4500000000	0.05000000000	0.5000000000		
4	*Össz.*	0.6777777778	0.3222222222	1.		
5						
6						

A táblázat belsejében lev értékek (citromsárga) az adott sor és oszlop címkéiben olvasható események együttes bekövetkezésének valószínségeit mutatják (pl. 0.272 annak esélye, hogy a választott doboz az *e l s* volt, *és* emellett *kék* szín golyó lett végül a választás). A táblázat bels értékeinek összege 1 (a teljes 100 % valószínség az együttes eloszlás valószínségei között oszlik meg). A sorok ill. oszlopok peremeken megjelen összegei (narancssárga) az adott dobozból való húzás ill. az adott szín golyó sorsolás események valószínségeit mutatják. A sorösszegek összegei ill. az oszlopösszegek összegei természetesen ugyancsak 1-et adnak (rózsaszín).

c) Mekkora a valószínsége, hogy egy sárga golyót az 1. illetve a 2. dobozból húztunk ki?

Egy ilyen, több ágon végigfuttatható kísérlet során a különböz ágak összesített valószínsége mellett a másik jellemz kérdés az lehet, hogy ha egy adott kimenet bekövetkezett, akkor amögött milyen valószínséggel történhetett adott ágon való lefutás. Pl. meg lehet kérdezni, hogy ha a példánkbeli kísérlet végeredményeként sárga golyó van a kezünkben, akkor az milyen valószínséggel származik egyik vagy másik dobozból. Ezt itt a feltételes valószínség definíciójával és az elz pontban kiszámolt $P(s)$ és $P(k)$

valószínségek felhasználásával határozzuk meg. Késbb viszont látni fogjuk majd, hogy pont az ilyen típusú kérdések gyors megválaszolására van kitalálva a *Bayes-tétel*.
Annak valószínségét, hogy az els dobozból *és* sárga golyót húzunk, a valószínségek *szorzás-tétel*ével határozzuk meg:

```
> psd1 := `P(s|d1)`*pd1;
```

Annak valószínsége, hogy egy kisorsolt sárga golyó az els dobozból érkezett (az els dobozon át sárga sorsolódás valószínsége arányítva a sárga sorsolódás teljes valószínségéhez):

```
> `P(d1|s)` := psd1/ps;
```

Annak valószínségét, hogy a második dobozból *és* sárga golyót húzunk, ismét a valószínségek szorzás-tételével határozhatjuk meg:

```
> psd2 := `P(s|d2)`*pd2;
```

Hasonlóan, annak valószínsége, hogy egy kisorsolt sárga golyó a második dobozból érkezett (a második dobozon át sárga sorsolódás valószínsége arányítva a sárga sorsolódás teljes valószínségéhez):

```
> `P(d2|s)` := psd2/ps;
```

Mivel sárga golyó kiválasztására összesen ez a két lehetség van, a két valószínség összege természetesen 1:

```
> `P(d1|s)` + `P(d2|s)`;
```

Látszik, hogy a sárga kisorsolásával végzd kísérletek nagyobb részében a húzott sárga golyó a második dobozból származik (kb. kétszer annyi esetben). Ez összhangban van azzal, hogy míg onnan szinte majdnem mindig sárga érkezik, az els dobozból csak az esetek nem egészen felében.

d) Milyen doboz-választási valószínségek mellett jutna érvényre a sorsolás végeredményében a golyók eredeti számaránya?

Bár a rendszerben a kék és sárga golyók száma egyenl, az adott kísérleti konstrukcióban túlnyomóan sárga golyók sorsolódnak ki. Ez azzal magyarázható, hogy az arányában sok sárgát, viszont darabszám szerint kevés golyót tartalmazó második doboz szerepe túlértékeldik azáltal, hogy kevés golyó mellett is ugyanazzal az 50 %-os kiválasztási valószínséggel rendelkezik, mint a 9-szer annyi golyót tartalmazó els doboz.
Szemléletünk azt sugallja, hogy a valódi számarányok jobban tükröznének, ha az els lépésben, a dobozok kiválasztásánál a valószínségek arányosak lennének a tartalmazott golyók számával: a több golyót tartalmazó doboz arányosan többször kerülne a sorsolásba.
Kíséreljük meg tehát a kiválasztási valószínségeinket ehhez igazítani:

```
> pd1 := d1/(d1 + d2);
```

Ebbl következen a másik doboz választásának valószínsége:

```
> pd2 := 1 - pd1;
```

Végezzük el ezekbl kiindulva újra a fentebbi számításokat.
A kék golyó kisorsolódásának teljes valószínsége:

```
> pk := `P(k|d1)`*pd1 + `P(k|d2)`*pd2;
```

Sárga golyó kisorsolódásának valószínsége:

```
> ps := `P(s|d1)`*pd1 + `P(s|d2)`*pd2;
```

Meggyzen látszik, hogy így a végeredményben már érvényre jut a golyók egyenl számaránya.

8. Feladat (Alkatrészek kivétele dobozból, visszatevés nélkül)

Egy dobozban 6 hibátlan és 4 hibás alkatrész van.

a) Ha egymás után véletlenszeren, visszatevés nélkül kiveszünk 3 alkatrészt a dobozból, mi a valószínsége annak, hogy az els kett hibátlan, a harmadik pedig hibás lesz?
b) Mekkora az esélye, hogy másodszorra hibátlant/hibásat húzunk ki (ha nem teszünk fel semmit az els húzás eredményérl)?
c) Függetlenek-e egymástól az 'elsre hibásat húzunk' és a 'másodszorra hibásat húzunk' események?

Megoldás

```
> restart;
```

Rögzítsük az összes, a hibátlan és a hibás alkatrészek számát!

```
> n := 10; # összes alkatrész
  g := 6; # hibátlanok száma
  h := 4; # hibásak száma
```

a) Ha egymás után véletlenszeren, visszatevés nélkül kiveszünk 3 alkatrészt a dobozból, mi a valószínsége annak, hogy az els kett hibátlan, a harmadik pedig hibás lesz?

Jelölje A_i azt az eseményt, hogy az i-edik alkalommal hibátlan alkatrészt veszünk ki ($1 \le i \le 3$). Ekkor a feladat megoldása a $P\left(A_1 \cdot A_2 \cdot \overline{A_3}\right)$ valószínség meghatározását jelenti, ami az általános szorzás-tétel szerint felírható $P\left(A_1\right) \cdot P\left(A_2 | A_1\right) \cdot P\left(\overline{A_3} | A_2 \cdot A_1\right)$ alakban.

Annak valószínsége, hogy elsre hibátlant húzunk:

```
> pA1 := g/n;
```

Mivel nem tesszük vissza a kihúzottat, ezért a dobozban már csak 9 alkatrész maradt, melybl 5 hibátlan. Így annak a valószínsége, hogy másodikra hibátlant húzunk, feltéve, hogy elsre is hibátlant húztunk:

```
> `P(A2|A1)` := (g - 1)/(n - 1);
```

Hasonlóan, a harmadik húzásra a dobozban már csak 8 alkatrész maradt, melybl 4 hibátlan és 4 hibás. Így annak a valószínsége, hogy hibásat húzunk, feltéve, hogy elsre és másodszorra is hibátlant húztunk:

```
> `P(A3c|A1A2)` := h/(n - 2);
```

Tehát a keresett valószínség:

```
> pA1A2A3c := pA1*`P(A2|A1)`*`P(A3c|A1A2)`;
```

Azaz, az esetek 6-odrészében következik be a vizsgált esemény.

b) Mekkora az esélye, hogy másodszorra hibátlant/hibásat húzunk ki (ha nem teszünk fel semmit az els húzás eredményérl)?

Az A_1 és $\overline{A_1}$ események teljes eseményrendszert alkotnak (a kett közül valamelyik biztosan bekövetkezik és egymást kölcsönösen kizárják), így használhatjuk a teljes valószínség tételét $P\left(A_2\right)$ ill. $P\left(\overline{A_2}\right)$ meghatározására:

$$P\left(A_2\right) = P\left(A_2 | A_1\right) \cdot P\left(A_1\right) + P\left(A_2 | \overline{A_1}\right) \cdot P\left(\overline{A_1}\right)$$

$$P\left(\overline{A_2}\right) = P\left(\overline{A_2} | A_1\right) \cdot P\left(A_1\right) + P\left(\overline{A_2} | \overline{A_1}\right) \cdot P\left(\overline{A_1}\right)$$

Elször számítsuk ki a hiányzó valószínségeket az a) pont gondolatmenetét követve:

```
> pA1c := h/n; # = 1 - pA1
> `P(A2|A1c)` := g/(n - 1);
> `P(A2c|A1)` := h/(n - 1);
```

```
> `P(A2c|A1c)` := (h - 1)/(n - 1);
```
A keresett valószínségeket a fenti képletekkel határozzuk meg.
```
> pA2 := `P(A2|A1)`*pA1 + `P(A2|A1c)`*pA1c;
  pA2c := `P(A2c|A1)`*pA1 + `P(A2c|A1c)`*pA1c;
```
Természetesen ezek összegének ki kell adnia 1-et:
```
> pA2 + pA2c;
```
Vegyük észre, hogy tulajdonképpen azt igazoltuk, amit amúgy is sejthettünk: elsre és másodszorra is ugyanakkora eséllyel akadhat a kezünkbe hibás alkatrész.
```
> evalb(pA1c = pA2c);
```
c) **Függetlenek-e egymástól az 'elsre hibásat húzunk' és a 'másodszorra hibásat húzunk' események?**

Vizsgáljuk $\overline{A}_1$ és $\overline{A}_2$ függetlenségét! Bár a b) feladatrészben láttuk, hogy a két esemény azonos valószínség, mégis úgy érezzük, hogy nem függetlenek egymástól, mert az els húzás eredménye befolyásolja a második valószínségét. Ellenrizzük a két esemény függetlenségére tanult $P(A|B) = P(A)$ feltétel teljesülését (ezt csak akkor használhatjuk jogosan, ha a feltétel-esemény pozitív valószínség, ami itt igaz)!
```
> `P(A2c|A1c)` = pA2c;
  fuggetlen = evalb(%);
```
Láthatjuk, hogy nem teljesül a feltétel, tehát $\overline{A}_1$ és $\overline{A}_2$ *nem független* események.

Másik megoldásként ellenrizhetnénk a függetlenség általános feltételét is: $P(A \cdot B) = P(A) \cdot P(B)$.

Ehhez $P\left(\overline{A}_1 \cdot \overline{A}_2\right)$-t a szorzás-tétellel számolhatjuk ki.
```
> pA1cA2c := `P(A2c|A1c)`*pA1c;
  pA1cA2c = pA1c*pA2c; # függetlenség feltétele
  fuggetlen = evalb(%);
```

9. Feladat (Vizsgázó hallgatók)

Egy évfolyam hallgatóinak 20%-a matematikából, 12%-a fizikából és 8%-a matematikából és fizikából is elégtelenre vizsgázott. Válasszunk ki egy hallgatót az évfolyamból és állapítsuk meg:

a) Mi a valószínsége annak, hogy matematikából elégtelen az osztályzata, ha fizikából elégtelen?

b) Mi a valószínsége annak, hogy matematikából vagy fizikából elégtelen az osztályzata?

c) Mi a valószínsége annak, hogy fizikából legalább elégséges az osztályzata, ha matematikából legalább elégséges?

Válaszoljuk meg továbbá, hogy

d) mekkora valószínséggel teljesítené sikerrel a hallgató mind a matematika, mind a fizika vizsgát, ha a két vizsga sikeres teljesítése független lenne egymástól (de továbbra is rendre 20 % illetve 12 % lenne az elégtelen osztályzat aránya)!

Megoldás
```
> restart;
```
Jelölje M ill. F azt az eseményt, hogy egy hallgató matematikából ill. fizikából elégtelenre vizsgázott. A feladat szövegébl kiolvashatjuk, hogy $P(M) = 0.2$, $P(F) = 0.12$ és $P(M \cdot F) = 0.08$.
```
> pM := 0.2;
  pF := 0.12;
  pMF := 0.08;
```

a) Mi a valószínsége annak, hogy matematikából elégtelen az osztályzata, ha fizikából elégtelen?
A $P(M|F)$ feltételes valószínséget keressük, melyet a feltételes valószínség definíciója alapján a $P(M|F)=\dfrac{P(MF)}{P(F)}$ képlettel számolhatunk ki:

```
> `P(M|F)` := pMF/pF;
```

Tehát a fizikából elégtelenre vizsgázó hallgatók $\dfrac{2}{3}$ -a matematikából is elégtelenre vizsgázik.

b) Mi a valószínsége annak, hogy matematikából vagy fizikából elégtelen az osztályzata?
Az események összegének valószínségére tanult képletet alkalmazzuk:
$P(M+F)=P(M)+P(F)-P(M{\cdot}F)$:

```
> `P(M + F)` := pM + pF - pMF;
```

Azt kaptuk, hogy a hallgatók 24 %-a bukott meg valamelyik tárgyból.

c) Mi a valószínsége annak, hogy fizikából legalább elégséges az osztályzata, ha matematikából legalább elégséges?
A 'legalább elégséges' az 'elégtelen' tagadása, így a $P(\overline{F}|\overline{M})$ feltételes valószínség a kérdés. Mivel a feltételes valószínség definíciója alapján $P(\overline{F}|\overline{M})=\dfrac{P(\overline{F}\cdot\overline{M})}{P(\overline{M})}$, ezért meg kell még határoznunk a $P(\overline{F}\cdot\overline{M})$ és a $P(\overline{M})$ valószínségeket.
Alkalmazzuk az ellentét esemény valószínségére tanult $P(\overline{A})=1-P(A)$ képletet és az $\overline{A}\cdot\overline{B}=\overline{A+B}$ de Morgan azonosságot:

$$P(\overline{M})=1-P(M)$$

$$P(\overline{F}\cdot\overline{M})=P(\overline{F+M})=1-P(M+F).$$

Szerencsére a $P(M+F)$ valószínséget már meghatároztuk a b) pontban.

```
> pMc := 1 - pM;
  pFcMc := 1 - `P(M + F)`;
> `P(Fc|Mc)` := pFcMc/pMc;
```

Tehát aki matematikából legalább elégségesre vizsgázott, az 95 %-os valószínséggel a fizika vizsgán is átment.

d) mekkora valószínséggel teljesítené sikerrel a hallgató mind a matematika, mind a fizika vizsgát, ha a két vizsga eredménye független lenne egymástól (de továbbra is rendre 20 % illetve 12 % lenne az elégtelen osztályzat aránya)!
Ha az $\overline{M}$ és $\overline{F}$ események függetlenek, akkor a kérdéses valószínség:
$P(\overline{M}\cdot\overline{F})=P(\overline{M})\cdot P(\overline{F})=P(\overline{M})\,(1-P(F))$.

```
> p_fuggetlen_McFc := pMc*(1 - pF);
```

Tehát 76 % helyett csak 70.4 % lenne az esélye annak, hogy a hallgató mindkét vizsgát sikerrel teljesíti. A hallgató szerencséjére azonban a két tárgyban elért sikeres eredmények között ers az összefüggés.

Gyakorló feladatok

Gy/1. Feladat (Áruház beszállítói / 1.)

Egy áruház három különböz beszállítótól veszi a banánt.

Az "A" beszállító termékeinek 20 %-a minségileg enyhén kifogásolható (másodosztályú), de mégis ettl veszik a teljes mennyiség 50 %-át, mert olcsó.
A "B" beszállító 95 %-ban els osztályú terméket szállít, de a többiekhez képest drágábban, ezért a teljes készlet 20 %-át veszik csak tlük.
A maradék a "C" beszállítótól származik, melynek termékei 10 %-ban másodosztályúak.
a) Mekkora valószínséggel származik egy termék a "C" beszállítótól? Mekkora hányada elsosztályú az "A", a "B" illetve a "C" beszállító termékeinek?
b) A bolt teljes kínálatában mennyi a kifogásolható termékek aránya?
c) Mekkora a valószínsége, hogy egy kiváló minség banán a legdrágább beszállítótól érkezik?

Megoldás

```
> restart;
```

Gy/2. Feladat (ZH, 2016, B csoport)

Határozzuk meg a $P(A)$, $P(B)$ és $P(A \cdot B)$ valószínségeket, ha adottak a $P(A|B) = \frac{2}{3}$, $P(A + B) = \frac{5}{6}$ és $P(\overline{A} \cdot B) = \frac{1}{6}$ valószínségek! Függetlenek-e az A és a B események?

Megoldás

```
> restart;
```

Gy/3. Feladat (Egér a konyhában)

Egy egér két lyukon tud bemenni a konyhába, onnan szintén két lyukon keresztül az éléskamrába, és a lyukak között mindig véletlenszeren választ. Mind a 4 lyuknál (egymástól függetlenül) 0.3 valószínséggel ül egy (egérre kiéhezett) macska. Feltéve, hogy az egér nem jutott be az éléskamrába, mennyi a valószínsége, hogy bejutott a konyhába?

Megoldás

```
> restart;
```